Especificaciones para la anotación de SPinTX con información lingüística para fines pedagógicos

Martí Quixal

Diciembre 2012 – hoy

Contents

Ι	Definición de LIST y SET	5
1	LIST y SET basados en información morfosintáctica 1.1 Categorías gramaticales	6 6
2	LIST y SET basados en lemas 2.1 Los LIST	7 7 7
3	LIST y SET basados en palabras (literales)	8
4	LIST y SET basados en etiquetas pedagógicas	9
5	Definiciones de templates	10
II 6	Secciones previas Reglas para rectificar errores del análisis morfosintático	11 13
II	I Secciones principales	17
7	Uso de los determinantes7.1Usos de los artículos definidos7.2Usos de los determinantes demostrativos7.3Usos pronominales de los demostrativos7.4Usos de los determinantes posesivos7.5Usos de los determinantes indefinidos	19 19 20 20 20
8	Género gramatical y artículos8.1 Género masculino en palabras terminadas en "a"	21 21 21
9	Uso de los adjetivos	22
10	Uso de las preposiciones 10.1 Usos de para	23 23

		10.1.2 El uso de <i>para</i> para expresar una finalidad	
		10.1.3 El uso de <i>para</i> para expresar una opinión	
	10.0	10.1.4 Usos de <i>para</i> no clasificados	
	10.2	Usos de por	
		10.2.1 Usos de <i>por</i> en locuciones	
		10.2.2 Usos de <i>por</i> no clasificados	
		10.2.3 Usos de preposiciones no clasificados	25
11		•	26
	11.1	Pronombres de sujeto: los pronombres de sujeto personales	26
		11.1.1 Pronombres de sujeto: <i>uno</i> como sujeto impersonal	
	11.2	Pronombres de objeto directo o complemento atributivo	27
	11.3	Pronombres de objeto directo, indirecto, o posesivos	27
	11.4	Pronombres de objeto indirecto	27
	11.5	Pronombres de objeto doble	27
	11.6	Pronombre se (solo)	28
19	Forr	nas verbales conjugadas	29
14		Formas verbales para tiempos de indicativo	
	12.1	12.1.1 Pretérito imperfecto	
			29
	19 9		20 30
	12.2		30
			30
			30
	19 3		30
	12.5	19.2.1 Usos do commendar	31
	19.4		32
			32 32
	12.5	12.5.1 Formas verbales para tiempos de imperativo informal (tú)	
			32 32
		12.5.2 Formas verbales para tiempos de imperativo iormai (usted)	
	10.6	Verbos copulativos	
	12.0		
		12.6.1 Verbo ser	33
	19.7		
	12.1		33
	10.0		33
	12.8	, ,	33
			33
			$\frac{34}{3}$
		12.8.3 Verbo ser en su uso como auxiliar	34
13			36
	13.1	1	36
		13.1.1 Expresar deseos con incertidumbre/en el futuro con agentes distintos en prin-	
			36
	13.2	<u> </u>	38
		13.2.1 Estilo indirecto	38

		13.2.2 Expresando opiniones	38
	13.3	Expresar opiniones, dudas	
	13.4	Expresar objetivos (con para que)	38
	13.5	Expresar deseos y recomendaciones (implicando a uno mismo)	38
			39
	14.1	Marcadores discursivos	36
		14.1.1 Marcadores discursivos indicadores de revisión	36
		14.1.2 Marcadores discursivos indicadores de causa	40
		14.1.3 Marcadores discursivos indicadores de igualdad	41
		14.1.4 Elementos discursivos usadors para ganar tiempo	42
15 :	Estr	ructuras características del habla en hablantes de herencia	43
	15.1	Estructuras derivadas del contacto entre lenguas	43
		15.1.1 Expresiones con cambio de código	43
	15.2	Usos de inidicativo dónde se esperaría subjuntivo (en español estándard)	43
			4 4
	16.1	Nombres propios	44
		16.1.1 Nombres de localizaciones geográficas	44
		16.1.2 Combinaciones de verbo y preposición de régimen	

Introducción

Este documento describe el proceso de anotación de las transcripciones de los clips del corpus SPinTX (Spanish in Texas) con información pedagógica para facilitar el uso de los mismos en la enseñanza y el aprendizaje de español como lengua extranjera.

DOCU



Part I Definición de LIST y SET

LIST y SET basados en información morfosintáctica

- 1.1 Categorías gramaticales
- 1.2 Rasgos morfosintácticos

LIST y SET basados en lemas

- 2.1 Los LIST
- 2.2 Los SET



LIST y SET basados en palabras (literales)



LIST y SET basados en etiquetas pedagógicas



Definiciones de templates

Los "templates" son un tipo de entidad especial en CG3 que sirve para expresar contextos o secuencias de palabras, lemas o etiquetas morfosintácticas que ocurren frecuentemente.



Part II Secciones previas



Reglas para rectificar errores del análisis morfosintático

Rule | SUBSTITUTE (Noun NC Singular Masculine) (Conjunction CC) TARGET ("pero");

- ETIQ: CC
- EJEM: pero
- DESC: Regla para cambiar a *pero* la lectura de nombre por la de conjunción. Ahora se aplica indiscriminadamente, pero podría ser conveniente controlar que antes del pero no hay un artículo. Por ejemplo: "el pero de la cuestión es que podría romperse el motor".

Rule

```
SUBSTITUTE ("hecho" Noun NC Singular Masculine) ("hacer" Verb VLadj)
TARGET ("hecho") IF (-1 Haber + Verbo);
```

- ETIQ: Cambia Noun NC Singular Masculine por Verb VLadj
- EJEM: hecho
- DESC: Regla para cambiar a *pero* la lectura de nombre por la de conjunción. Ahora se aplica indiscriminadamente, pero podría ser conveniente controlar que antes del pero no hay un artículo. Por ejemplo: "el pero de la cuestión es que podría romperse el motor".

Rule

```
SUBSTITUTE ("ranchar" Verb VLfin Pres Indi Sing 1st) ("rancho" Noun NC Singular Masculine) TARGET ("ranchar");
```

- ETIQ: Cambia Verb VLfin Pres Indi Sing 1st por Noun NC Singular Masculine en la palabra "rancho".
- EJEM: rancho
- DESC: Lo hace de forma indiscriminada. Quizá en el futuro habrá que cambiarla, refinarla.

Rule

```
SUBSTITUTE ("dicho" Determiner QU) ("decir" Verbo VLadj) TARGET ("dicho");
```

• ETIQ: Cambia lectura de "dicho" como determinante por lectura como adjetivo/participio.

- EJEM: dicho
- DESC: Lo hace de forma indiscriminada. Quizá en el futuro habrá que cambiarla, refinarla.

SUBSTITUTE ("oír" Verb VLfin Pres Subj Sing 1st_or_3rd) ("oír" Verb VLfin Pres Indi_Command Sing 2nd_3rd) TARGET ("oír");

- ETIQ: Cambia lectura de "oye" como presente de subjuntivo por lectura de presente de indicativo/imperativo.
- EJEM: oye
- DESC: Lo hace de forma indiscriminada. Quizá en el futuro habrá que cambiarla, refinarla.

Rule

SUBSTITUTE ("ser" Verb VSfin Pret Indi Plural 3rd) ("ir" Verb VLfin Pret Indi Plural 3rd) TARGET ("<fueron>") IF (-1C SeLiteral);

Rule

SUBSTITUTE ("ser" Verb VSfin Pret Indi Plural 3rd) ("ir" Verb VLfin Pret Indi Plural 3rd) TARGET ("<fueron>") IF (1 ConLema + Prep);

Rule

SUBSTITUTE ("ser" Verb VSfin Pret Indi Plural 3rd) ("ir" Verb VLfin Pret Indi Plural 3rd) TARGET ("<fueron>") IF (1 Gerundio);

- ETIQ: Reglas que cambian la lectura de "ser" por "ir" en "fueron" cuando no va precedida de un pronombre se o seguida la preposición con o un gerundio.
- EJEM: se fueron juntos, se fueron a la escuela
- DESC: Quizá en el futuro habrá que cambiarlas, refinarlas.

Rule

SUBSTITUTE ("maestro" Adjective ADJ Singular) ("maestro" Noun Singular) TARGET ("maestro");

Rule

SUBSTITUTE ("maestro" Adjective ADJ Plural) ("maestro" Noun Plural) TARGET ("maestro");

- ETIQ: Cambia lectura de "maestro" de adjetivo sustantivo.
- EJEM: maestro, maestra, maestros, maestras
- DESC: Lo hace de forma indiscriminada. Quizá en el futuro habrá que cambiarla, refinarla. Por ejemplo, en "obras maestras", que no ocurre en la versión actal del corpus.

Rule

SUBSTITUTE ("erar|ser" Verb VLfin Past Subj Sing 2nd) ("ser" Verb VSfin Imp Indi Sing 2nd) TARGET ErasLiteral;

SUBSTITUTE ("erar|ser" Verb VLfin Past Subj Plural 2nd_or_3rd) ("ser" Verb VSfin Imp Indi Plural 3rd) TARGET EranLiteral;

- ETIQ: Cambia lectura de "eras" y "eran" de lema ambiguo entre erar y ser a sólo lema ser.
- EJEM: eras, eran
- DESC: Consideramos erar un verbo muy poco frecuente. Si hiciera falta, se podría refinar la regla.

Rule

SUBSTITUTE ("UNK" Noun NC Plural) ("mío" Pronoun PPO Plural) TARGET ("<mis>") OR ("<Mis>");

Rule

SUBSTITUTE ("UNK" Noun NC Plural) ("nuestro" Pronoun PPO Plural) TARGET ("<nuestros>") OR ("<Nuestros>");

- ETIQ: Cambia lectura de "UNK" Noun NC Plural a "mío" Pronoun PPO Plural a "mis".
- EJEM: mis, Mis
- DESC: Lo hacemos con nuestros (y quizás habrá que añadir otros posesivos).

Rule

SUBSTITUTE ("dar|decir" Verb VLfin Pres Command Sing 2nd)
("dar" Verb VLfin Pret Indi Sing 1st) TARGET DiLiteral IF (-1
PronombresCliticosTODOS);

- ETIQ: Cambia lectura de "di" de lema ambiguo entre dar y decir a sólo lema dar.
- EJEM: le di, me di cuenta
- DESC: Dado que "di" como mandato llevará siempre el clítico pospuesto, si no lo lleva lo podemos considerar una forma de dar. Cabría tener en cuenta otras opciones pero esta parece razonable.

Rule

```
SUBSTITUTE (Verb VLinf Pres Command Plural 2nd) (Verb VLinf) TARGET ("<(.*)(ar|er|ir|ír)>"r);
```

- ETIQ: Cambia lectura Command Plural 2nd a los infinitivos marcados como imperativo (y que estrictamente no los son)
- EJEM: decir, nadar, temer
- DESC: Se aplica a cualquier infinitivo con la lectura arriba mencionada. A pesar de que los imperativos pueden tener esta función de mandatopragmáticamente hablando, consideramos que no se deben etiquetar morfológicamente como tales.

OBSERVACIÓN: Notad la "r" al final del TARGET en la regla. Indica que lo anterior debe ser tratado como una expresión regular. Si no, la regla no se aplica.

SUBSTITUTE ("UNK" Verb VLfin Pres Command Plural 2nd) ("platicar" Verb VLfin Pres Command Sing 3rd) TARGET ("<platiquenos>");

- ETIQ: Cambia lectura Command Plural 2nd y lema a "platíquenos"
- EJEM: platíquenos
- DESC: Se aplica sólo a "platíquenos".

Rule

SUBSTITUTE ("tener" Verb VMfin) ("tener" Verb VMfin Past Subj Plural 3rd) TARGET ("<tengan>");

- ETIQ: Añade lectura de pasado de subjuntivo a tener cuando este recibe el tag VMFin.
- EJEM: tengan
- DESC: Se aplica sólo a "tengan".

Rule

SUBSTITUTE ("para" Conjunction CSUBI) ("para" Preposition PREP) TARGET ("para");

- ETIQ: Cambia la lectura de conjunción de subordinación de infinitivas por la de preposición.
- EJEM: tengan
- DESC: Se aplica sólo a "para" como lema.

OBSERVACIÓN: Aunque es verdad que "para" puede ir seguida de un infinitivo para formar una subordinada infinitiva esta es una función que pueden desempeñar otras preposiciones, y creo que no por ello hay que cambiarles su categoría gramatical principal.

Rule

SUBSTITUTE ("crear" Verb VLfin Past Subj Plural 3rd) ("creer" Verb VLfin Pres Indi Plural 3rd) TARGET ("<creen>") IF (1 LoLiteral) (2 MismoLiteral) (3 Que);

Rule

SUBSTITUTE ("crear" Verb VLfin Past Subj Plural 3rd) ("creer" Verb VLfin Pres Indi Plural 3rd) TARGET ("<creen>") IF (1 En OR Que);

Rule

SUBSTITUTE ("crear" Verb VLfin Past Subj Plural 3rd) ("creer" Verb VLfin Pres Indi Plural 3rd) TARGET ("<creen>") IF (-2 NoAdv) (-1 PronombresCliticosTODOS OR SeLiteral);

- ETIQ: Varias reglas para dar la lectura indicativa de "creer" a la forma "creen" en lugar de la subjuntiva de "crear".
- EJEM: tengan
- DESC: Se aplica teniendo en cuenta algunos contextos que generalmente decantaran la lectura hacia la interpretación prevista (creer).

Part III Secciones principales

Las reglas de identificación de fenómenmos pedagógicos están ordenadas de más concretas a más generales. Así, por ejemplo, se detectan primero, cuando es posible, los distintos usos de para (opinión, intención, destino, etc. como @Gram:Prep:Para:Opinion...) y luego se marcan como simplemente @Gram:Prep:Para el resto de ocurrencias de para. Luego, se hace lo propio con otras preposiciones, hasta el punto que si para una preposición no hay reglas especiales, sus ocurrencias se marcan simplemente como @Gram:Prep.

Esta estrategia se utiliza para evitar que una misma palabra reciba por partida doble informaciones de igual granularidad. Para ello, antes de asignar una etiqueta, se comprueba que no haya otra asignada del mismo nivel. A efectos prácticos, etiquetas como @Gram:Prep:Para:Opinion...) y @Gram:Prep:Para se consideran del mismo nivel.

Para evitar este tipo de duplicidades también se podría usar una estrategia basada en la distinción entre ADD y MAP (véase la documentación de vislcg3). Por ahora no se ha hecho, porque esto impide la asignación de múltiples etiquetas a un mismo token/una misma palabra.

REM: Ahora mismo sólo estamos asignando hasta tres niveles de descripción pedagógica. Veáse el Google Doc Pedagogical tipologies https://docs.google.com/spreadsheet/ccc?key=OAlyBsKogCRnwdDd3ZWRjejZlQU5XZWl6WVVUZmtKcEE&usp=sharing.



Uso de los determinantes

7.1 Usos de los artículos definidos

Identificaremos los usos de los artículos definidos el, la, los, las, y el artículo (en algunas gramáticas considerado pronombre) lo cuando va seguido de un adjetivo.

Rule

```
ADD (@Gram:Det:Art) Articulo IF (O DetArtDefinidosLiterales) (NOT 1 FiniteVerbForms OR PronombreDeObjetoNoAmbiguos OR SeLiteral);
```

- ETIQ: Asigna @Gram:Det:Art
- EJEM: el, la, los, las
- DESC: Cualquier ocurrencia no seguida de un verbo finito o un pronombre clítico, porque treetagger no distingue entre su lectura como determinante y la pronominal.

Rule

```
ADD (@Gram:Det:Art) Articulo IF (O LoLiteral) (*1 Adj BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);
```

- ETIQ: Asigna @Gram:Det:Art
- EJEM: lo + ADJETIVO
- DESC: Cualquier ocurrencia de LO seguida de un adjetivo (con un adv opcional en medio).

7.2 Usos de los determinantes demostrativos

Rule

```
ADD (@Gram:Det:Demo) ArtPronDemo IF (-1 Prep OR FiniteVerbForms) (0 EsteEseAquel) (NOT O EstoEsoAquello) (1 Nombre);
```

- ETIQ: Asigna @Gram:Det:Demo
- EJEM: Prep/Verbo este/ese/aquel (LEMMA) + Nombre
- DESC: Cualquier ocurrencia de un demostrativo precedida de una preposición o un verbo conjugado y seguido de un nombre.

7.3 Usos pronominales de los demostrativos

Rule ADD (@Gram:Det:Demo) ArtPronDemo IF (-1 Prep OR FiniteVerbForms) (0 EstoEsoAquello);

- ETIQ: Asigna @Gram:Pron:Demo
- EJEM: Prep/Verbo esto/eso/aquello
- DESC: Cualquier ocurrencia de un demostrativo terminado en -o. No distinguimos las formas que forman parte de locuciones (como "por eso", etc.).

7.4 Usos de los determinantes posesivos

Rule ADD (@Gram:Det:Pos) DetPosSing IF (*1 Nombre + SG BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

Rule ADD (@Gram:Det:Pos) DetPosPlu IF (*1 Nombre + PL BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

Rule | ADD (@Gram:Det:Pos) DetPosFemPlu IF (*1 Nombre + PL BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

Rule | ADD (@Gram:Det:Pos) DetPosMascPlu IF (*1 Nombre + PL BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

Rule | ADD (@Gram:Det:Pos) DetPosFemSing IF (*1 Nombre + SG BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

Rule ADD (@Gram:Det:Pos) DetPosMascSing IF (*1 Nombre + SG BARRIER FiniteVerbForms OR Conjuncion OR LimiteOracionSimple);

- ETIQ: Asigna @Gram:Det:Poss
- EJEM: Prep/Verbo mi/tu/su/sus/nuestro/nuestros/vuestro/vuestros (LEMMA) + Nombre
- DESC: Cualquier ocurrencia de un determinante posesivo seguido de un nombre (se admiten ADV y ADJ intercalados).

7.5 Usos de los determinantes indefinidos

Rule ADD (@Gram:Det:CuantIndef) UnoLiteral IF (1 De) (NOT -1 Infinitive);

- ETIQ: Asigna @Gram:Det:CuantIndef
- EJEM: Que no los puedo criticar porque yo soy uno de ellos (...)
- DESC: Cualquier ocurrencia de un determinante posesivo seguido de un nombre (se admiten ADV y ADJ intercalados).

Género gramatical y artículos

8.1 Género masculino en palabras terminadas en "a"

Rule

```
ADDRELATION (Gram:Genero:ExcepMasc) DetAllIncluded IF (0 ElDetLiteral + SG OR UnDetLiteral + SG OR ArtPronDemo + SG) TO (1 SustMascAcabadosEnA + Nombre + SG);
```

- ETIQ: Asigna R:Gram:Genero:ExcepMasc
- EJEM: el/un/ese + (SUST acabado en -a, tema, día, problema, trauma)
- DESC: No se aplica con todos los determinantes porque TreeTagger no asigna género a todos y lematiza con criterios curiosos.

8.2 Cambio de artículo femenino a masculina en palabras con "a" tónica inicial

Rule

```
ADDRELATION (Gram:Genero:ExcepArtFem) Articulo IF (0 ElDetLiteral) TO (1 SustFemConATonicaInicial + Nombre + SG);
```

- ETIQ: Asigna R:Gram:Genero:ExcepArtFem
- EJEM: el + (SUST con a- tónica inicial, arma, aula, agua)
- DESC: No se aplica con todos los determinantes porque TreeTagger no asigna género a todos y lematiza con criterios curiosos.

Uso de los adjetivos

Rule

ADD (Gram:Adj:Ord) AdjOrd IF (O AdjOrd) (NOT O LiteralPrimeroOrTercero + SG) (-1 DetAllIncluded + SG OR Prep OR PrepContract OR Verbo) (1 Sustantivo + SG);

Rule

ADD (Gram:Adj:Ord) AdjOrd IF (O AdjOrd) (NOT O LiteralPrimeroOrTercero + PL) (-1 DetAllIncluded + PL OR Prep OR Verbo) (1 Sustantivo + PL);

- ETIQ: Asigna Gram:Adj:Ord
- EJEM: por primera vez, del tercer mundo
- DESC: Se controla que el elemento de la izquierda sea un determinante, preposición, con o sin contracción, o verbo. El de la derecha un sustantivo y, cuando hay número, deben concordar. PrepContract no tiene sentido en la versión plural porque no se hace contracción entre preposición y artículos en plural.

Uso de las preposiciones

10.1 Usos de para

Veáse http://dev.coerll.utexas.edu/spanishtx/spe/html/adv11.html.

10.1.1 El uso de para para expresar el destinatario o audiencia de una acción

Rule ADD (@Gram:Prep:Para) Prep IF (-1 SerOServir + Verbo) (0 Para) (NOT 0 GramPrep) (1 DetAllIncluded) (2 Personas);

- ETIQ: Asigna @Gram:Prep:Para
- EJEM: Fue escrito para los estudiantes.
- EJEM: Es fundamental para una persona.
- DESC: Requiere para + DET + NOMBRE < +humano >

```
Rule ADD (@Gram:Prep:Para) Prep IF (-1 SerOServir + Verbo) (0 Para) (NOT 0 GramPrep) (1 Personas);
```

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: Ropa para niños.
- DESC: Requiere para + NOMBRE < +humano >

10.1.2 El uso de *para* para expresar una finalidad

OBSERVACIÓN: Dudo de si el uso de "Juan es muy bueno para el fútbol" se parece o no al de "hemos venido antes para no quedarnos sin asiento". De momento los pongo juntos.

```
Rule ADD (@Gram:Prep:Para) Prep IF (0 Para) (NOT 0 GramPrep) (1 Infinitive);
```

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: Soy muy mala para recordar.
- DESC: Requiere para + INFINITIVO

ADD (@Gram:Prep:Para) Prep IF (0 Para) (NOT 0 GramPrep) (1 PronPersFuerte OR No) (2 Infinitive);

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: Había que saber hablar inglés para uno desarrollar su trabajo.
- DESC: Requiere para + PRONOMBRE PERSONAL NOMINATIVO + INFINITIVO

Rule ADD (@Gram:Prep:Para) Prep IF (0 Para) (NOT 0 GramPrep) (1 Que);

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: Me daba clases para que aprendiera inglés.
- DESC: Requiere para + QUE

10.1.3 El uso de para para expresar una opinión

Rule

ADD (@Gram:Prep:Para) Prep IF (0 Para) (NOT 0 GramPrep) (1 PronombrePersonal) (NOT 1 YoLiteral OR TuLiteral);

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: Para mi.
- DESC: Requiere para + PRONOMBRE PERSONAL ACUSATIVO
- C: Esta regla es muy floja. Habrá que revisarla.

10.1.4 Usos de para no clasificados

Rule ADD (@Gram:Prep:Para) Prep IF (0 Para) (NOT 0 GramPrep);

- ETIQ: Asigna @Gram:Prep:Para.
- EJEM: para (cualquier otra ocurrencia)
- DESC: Requiere para

10.2 Usos de por

Veáse http://dev.coerll.utexas.edu/spanishtx/spe/html/adv11.html.

10.2.1 Usos de *por* en locuciones

Rule

```
ADD (@Gram:Prep:Por) Prep IF (0 Por) (NOT 0 GramPrep) (1 LemasDePalabrasEnLocucionesConPor);
```

- ETIQ: Asigna @Gram:Prep:Por
- EJEM: por ejemplo, por eso, por tanto
- DESC: Requiere por seguida por alguna palabra en la lista de palabras predefinidas Lemas-DePalabrasEnLocucionesConPor

ADD (@Gram:Prep:Por) Prep IF (0 Por) (NOT 0 GramPrep) (1 LoLiteral) (2 LemasDePalabrasEnLocucionesConPor);

- ETIQ: Asigna @Gram:Prep:Por.
- EJEM: por lo tanto, por lo regular
- DESC: Requiere por seguida por LO y alguna palabra en la lista de palabras predefinidas LemasDePalabrasEnLocucionesConPor

10.2.2 Usos de por no clasificados

Rule | ADD (@Gram:Prep:Por) Prep IF (0 Por) (NOT 0 GramPrep);

- ETIQ: Asigna @Gram:Prep:Por.
- EJEM: por (cualquier ocurrencia)
- \bullet DESC: Requiere por

10.2.3 Usos de preposiciones no clasificados

Rule | ADD (@Gram:Prep) Prep (NOT 0 GramPrep);

- ETIQ: Asigna @Gram:Prep.
- EJEM: PREP (cualquier ocurrencia)
- \bullet DESC: Requiere PREP

Uso de los pronombres

11.1 Pronombres de sujeto: los pronombres de sujeto personales

Rule | ADD (@Gram:Pron:Sujeto) PronPersFuerte;

- ETIQ: Asigna @Gram:Pron:Sujeto.
- EJEM: yo, tú, él, ella, ud., nosotros, vosotros, ellos, ellas, uds. (cualquier ocurrencia)
- EJEM: Cuando aquí en los Estados Unidos uno creció con las lavadoras, ¿verdad?
- DESC: La ocurrencia de uno de ellos.

11.1.1 Pronombres de sujeto: uno como sujeto impersonal

Rule | ADD (@Gram:Pron:Sujeto) UnoLiteral IF (1 FiniteVerbForms);

Rule | ADD (@Gram:Pron:Sujeto) UnoLiteral IF (-2 NoAdv OR PronombresCliticosTODOS OR Adverbio) (-1 Verbo);

Rule ADD (@Gram:Pron:Sujeto) UnoLiteral IF (*1 Adverbio BARRIER GrupoNominal OR LimiteOracionSimple LINK 1 FiniteVerbForms);

Rule ADD (@Gram:Pron:Sujeto) UnoLiteral IF (*1 Adverbio BARRIER GrupoNominal OR LimiteOracionSimple LINK 1 SeLiteral LINK 1 FiniteVerbForms);

- ETIQ: Asigna @Gram:Pron:Sujeto.
- EJEM: Cuando aquí en los Estados Unidos uno creció con las lavadoras, ¿verdad?
- EJEM: (...) uno ya sabe si hablan el español
- DESC: La ocurrencia de uno de ellos. En cuanto a "uno", sólo se aplica si va seguido de un verbo conjugado, o si va precedido del adverbio "no" o un clítico y un verbo (conjugado o no).

11.2 Pronombres de objeto directo o complemento atributivo

Rule ADD (@Gram:Pron:Objeto) ElLema IF (NOT -1 Prep) (O Articulo) (1 FiniteVerbForms);

- ETIQ: Asigna @Gram:Pron:Objeto
- EJEM: mí nunca me la hicieron pero que estabamos
- DESC: lo, la, los, las VERBO CONJUGADO

11.3 Pronombres de objeto directo, indirecto, o posesivos

Rule | ADD (@Gram:Pron:Objeto) PronPers IF (O PronombreDeObjetoNoAmbiguos);

- ETIQ: Asigna @Gram:Pron:Objeto
- EJEM: me, te, os, nos (cualquier ocurrencia)
- DESC: La ocurrencia de uno de ellos.

11.4 Pronombres de objeto indirecto

Rule | ADD (@Gram:Pron:Objeto) PronPersClitico IF (O LeLema);

- ETIQ: Asigna @Gram:Pron:Objeto
- EJEM: le, les
- DESC: La ocurrencia de uno de ellos.

11.5 Pronombres de objeto doble

Rule | ADDRELATION (Gram:Pron:ObjetoDoble) PronPers (O PronombreDeObjetoNoAmbiguos) TO (1 ElLema + Articulo);

Rule ADDRELATION (Gram:Pron:ObjetoDoble) PronPersSE (O SeLiteral) TO (1 ElLema + Articulo);

- ETIQ: Asigna @Gram:Pron:ObjetoDoble.
- EJEM: se lo, se la, se los, se las, me lo, me la, me los, me las, te lo, te la, te los, te las, etc.
- DESC: La ocurrencia de cualquier combinación de dos pronombres clíticos.

11.6 Pronombre se (solo)

 \mathbf{Rule}

ADD (@Gram:Pron:Otros) PronPersSE (O SeLiteral) (NOT O GramPron) (NOT 1 FiniteVerbForms);

- ETIQ: Asigna @Gram:Pron:Otros.
- EJEM: se lo, se la, se los, se las, me lo, me la, me los, me las, te lo, te la, te los, te las, etc.
- DESC: La ocurrencia de cualquier combinación de dos pronombres clíticos.



Formas verbales conjugadas

12.1 Formas verbales para tiempos de indicativo

12.1.1 Pretérito imperfecto

Rule

```
ADD (@Gram:Imperf) FiniteVerbForms IF (0 FiniteVerbForms + PretImperfSimple) (NOT 0 Haber);
```

- ETIQ: Asigna @Gram:Imperf.
- EJEM: cantaba, corría, temía, partía (cualquier ocurrencia)
- DESC: Requiere una forma identificada como pretérito simple que no sea del verbo haber (auxiliar).

12.1.2 Pretérito perfecto

Pretérito perfecto simple

Rule

```
ADD (@Gram:Indefinido) FiniteVerbForms IF (0 FiniteVerbForms +
PretPerfSimple) (NOT 0 Haber);
```

- ETIQ: Asigna @Gram:PretPerf.
- EJEM: cantó, corrió, anduvo, partió (cualquier ocurrencia)
- DESC: Requiere una forma identificada como pretérito perfecto simple que no sea del verbo haber (auxiliar).

Pretérito perfecto perifrástico o compuesto

```
Rule
```

```
ADDRELATION (Gram:PerfPerif Gram:Aux:Haber) FiniteVerbForms (0 Haber + FiniteVerbForms + Presente) (NOT 0 Subjuntivo) TO (1 Part);
```

- ETIQ: Asigna R:Gram:PerfPerif a la forma del verbo HABER
- EJEM: ha cantado, ha corrido, ha andado dormido (cualquier ocurrencia)
- DESC: Requiere una forma finita de HABER seguida de un participio

12.2 Verbos piscológicos: gustar, encantar, sorprender y otros

Veáse http://dev.coerll.utexas.edu/spanishtx/spe/html/adv13.html. Veáse http://dev.coerll.utexas.edu/spanishtx/spe/html/beg05.html.

12.2.1 Usos de *gustar* para expresar gustos propios

Rule ADDRELATION (Gram:Gustar:Propios) PronPers (O Me OR NosLiteral) TO (1 Gustar-Like + 3aPersAmbig);

- ETIQ: Asigna @Gram:Gustar:Propios.
- EJEM: me gusta ayudar a mis amigos
- DESC: Requiere la forma ME seguida de una forma de GUSTAR en tercera persona (posiblemente ambigua)

Rule ADDRELATION (Gram:Gustar:Propios) PronPers (0 Me OR NosLiteral) (1 Haber) TO (1 Gustar-Like + Part);

- ETIQ: Asigna R:Gram:Gustar: Propios a la forma del verbo HABER
- $\bullet\,$ EJEM: $me\ ha\ gustado$ ayudar a mis amigos
- EJEM: me hubieras gustado con un vestido verde
- DESC: Requiere la forma ME seguida de una forma de HABER en tercera persona (posiblemente ambigua)

12.2.2 Usos de gustar para expresar gustos de tu interlocutor

Rule | ADD (@Gram:Gustar:Interlocutor) FiniteVerbForms IF (-1 Tu) (0 Gustar-Like + 3aPersAmbig);

- ETIQ: @Gram:Gustar:Interlocutor
- EJEM: Dime, te gusta vivir en Estados Unidos.]
- DESC: Secuencia del lema TE y un forma de GUSTAR en tercera persona del singular.

12.2.3 Usos de *gustar* para expresar gustos de otras personas

Rule ADDRELATION (Gram:Gustar:Ajenos) LeLema (O LeLema) TO (1 Gustar-Like + 3aPersAmbig);

- ETIQ: R:Gram:Gustar:Ajenos a la forma del verbo GUSTAR
- EJEM: Dime, le gusta vivir en Estados Unidosa tu hijo.
- DESC: Secuencia del lema LE y un forma de GUSTAR en tercera persona del singular.

12.3 Expresar sorpresa

Veáse http://dev.coerll.utexas.edu/spanishtx/spe/html/adv13.html.

12.3.1 Usos de sorprender

Rule

ADDRELATION (Gram:Gustar:Otros) Yo (0 Yo) TO (1 Sorprender-Like + 3aPersAmbig);

Rule

ADDRELATION (Gram:Gustar:Otros) Yo (O Yo) To (*1 Sorprender-Like BARRIER GrupoNominal OR LimiteOracionSimple LINK -1 Haber + 3aPersAmbig);

- ETIQ: Asigna Gram:Gustar:Otros (Gram:ExpresarSorpresas:Propias) a la forma de SOR-PRENDER
- EJEM: Siempre me sorprende su forma de hablar.
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Gram:Gustar:Otros) Tu (0 Tu) To (1 Sorprender-Like + 3aPersAmbig);

- ETIQ: Asigna Gram:Gustar:Otros (antes era Gram:ExpresarSorpresas:Interlocutor) a la forma de SORPRENDER
- EJEM: ¿Y te sorprende su forma de hablar?
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Gram:Gustar:Otros) Tu (O Tu) To (*1 Sorprender-Like BARRIER GrupoNominal OR LimiteOracionSimple LINK -1 Haber + 3aPersAmbig);

- ETIQ: Asigna Gram:Gustar:Otros (antes era Gram:ExpresarSorpresas:Interlocutor) a la forma de HABER
- EJEM: ¿Hay alguna palabra mexicana que te haya sorprendido?
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Gram:Gustar:Otros) LeLema (O LeLema) To (1 Sorprender-Like + 3aPersAmbig);

- ETIQ: Asigna Gram:Gustar:Otros (antes era Gram:ExpresarSorpresas:Ajenas) a la forma de SORPRENDER
- EJEM: ¿Y le sorprende su forma de hablar?
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Gram:Gustar:Otros) LeLema (O LeLema) TO (*1 Sorprender-Like BARRIER GrupoNominal OR LimiteOracionSimple LINK -1 Haber + 3aPersAmbig);

- ETIQ: Asigna Gram:Gustar:Otros (antes era Gram:ExpresarSorpresas:Ajenas) a la forma de HABER
- EJEM: ¿Hay alguna palabra mexicana que le haya sorprendido?
- DESC: [Espacio para describir con palabras lo que hace la regla.]

12.4 Formas verbales para tiempos de subjuntivo

Rule ADD (@Gram:MSub:Pres) Verbo IF (0 Presente + Subjuntivo) (NOT -1 De) (NOT 1 Part);

Rule | ADD (@Gram:MSub:Pret) FiniteVerbForms IF (0 Verbo + Pasado + Subjuntivo) (NOT -1 De) (NOT 1 Part);

Rule ADDRELATION (Gram: MSub: Pres) FiniteVerbForms (O Haber + Verbo + Presente + Subjuntivo) (NOT -1 De) TO (1 Part);

Rule | ADDRELATION (Gram:MSub:Pret) FiniteVerbForms (O Haber + Verbo + Pasado + Subjuntivo) (NOT -1 De) TO (1 Part);

- ETIQ: Asigna Gram:MSub:Pres a verbos en presente o pretérito de subjuntivo
- EJEM: ¡Hay alguna palabra mexicana que le haya sorprendido?
- DESC: No se aplica si la forma verbal no esta precedida por una "de" para evitar formas incorrectamente marcadas como subjuntivo por el tagging básico. Si se trata de una forma compuesta, es decir, HABER + PART, se marca todo como tal.

12.5 Formas verbales para tiempos de imperativo

12.5.1 Formas verbales para tiempos de imperativo informal (tú)

Rule ADD (@Gram:MImp:Informal) Verbo IF (0 Presente + Imperativo + SG + 2aPers) (NOT 0 ("<órale>") OR ("<(.*)(é|í|á)rselo>"r));

- ETIQ: @Gram:MImp:Informal
- EJEM: cuéntame...
- DESC: Cualquier forma de imperativo de la segunda persona del singular. No se aplicará en formas como cantárselo, comérselo o partírselo, pero sí en otras formas infinitivas con pronombres de objeto que TreeTagger etiquete como imperativos.

OBSERVACIÓN: Actualmente falla en "díael" (debido a un error de transcripción). Se ha arreglado el problema de transcripción y en próximas versiones no debería aparecer.

12.5.2 Formas verbales para tiempos de imperativo formal (usted)

Rule | ADD (@Gram:MImp:Formal) Verbo IF (0 Presente + Imperativo + SG + 3aPers); OBSERVACIÓN: Sólo aparecen tres ocurrencias (detectadas) de uso del imperativo en la forma formal.

12.5.3 Formas verbales para tiempos de imperativo en negativo (no mires!)

No hemos anotado usos de imperativo en negativo (p.e., ¡No compre(s) un vino tan caro!) ni en el registro formal ni el informal.

12.6 Verbos copulativos

12.6.1 Verbo ser

Rule ADD (@Gram:SerEstar:Ser) Verbo (O VerboSer OR VerboSerAmbiguoConErar) (NOT -1 Haber);

Rule | ADD (@Gram:SerEstar:Ser) Verbo (O VerboSerAmbiguoConIr) (NOT 1 Prep) (NOT -1 Haber);

Rule ADDRELATION (Gram:SerEstar:Ser) Verbo (O Haber + FiniteVerbForms) TO (1 VerboSer + Part OR VerboSerAmbiguoConErar + Part) (NOT 1 Part);

- ETIQ: @Gram:SerEstar:Ser
- EJEM: soy, eres, ha sido,, pero no ha sido compradoetc.
- EJEM: fui, fuimos... NOT Prep
- DESC: Cualquier forma de ser del pretérito indefinido no seguido de una preposición (porque entonces seguramente será del verbo ir). También las formas perifrásticas con haber que no vayan seguidas de un participicio.

12.6.2 Verbo estar

Rule | ADD (@Gram:SerEstar:Estar) Verbo (0 VerboEstar);

- ETIQ: @Gram:SerEstar:Estar
- EJEM: estoy, estás, estaban, etc.
- DESC: Cualquier forma de estar

12.7 Verbo haber

12.7.1 Verbo haber en su uso com existencial

Rule ADD (@Gram:Haber:Exist) Verbo (0 Haber) (NOT 1 Part);

- ETIQ: @Gram:Haber:Exist
- EJEM: *Había* tres personas en el ascensor.
- DESC: Cualquier forma de haber que no vaya seguida de un participio

12.8 Verbos auxiliares: ser, estar y haber

12.8.1 Verbo estar en su uso como auxiliar

Se etiqueta con RGram:SerEstar:EstarAux las secuencias de estar seguido de un gerundio.

ADDRELATION (Gram:SerEstar:EstarAux) VerboEstar IF (O FiniteVerbForms) TO (1 Gerundio);

Rule

ADDRELATION (Gram:SerEstar:EstarAux) VerboEstar IF (0 Infinitive) TO (1 Gerundio);

- ETIQ: R:Gram:SerEstar:EstarAux
- EJEM: y me estuvo platicando allí
- DESC: Actualmente sólo se aplica sobre formas conjugadas y en secuencias estar (infinitivo) más gerundio. Habrá que ver cuán necesario es aplicar reglas para "hemos estado mirando" o "estando comprando?" (el interrogante los uso para marcar que dudo de la gramaticalidad de esta oración.

12.8.2 Verbo haber en su uso como auxiliar en tiempos verbales compuestos

Se etiqueta con @Gram:Aux:Haber, pero se hace en las reglas que sirven para etiquetar los tiempos verbales correspondientes. Véase por ejemplo la anotación (y la regla) del Presente Perfecto en la sección 12.1.2.

12.8.3 Verbo ser en su uso como auxiliar

No se etiquetan las ocurrencias de ser como auxiliar en las construcciones pasivas por dos razones: una es que hay muy pocas ocurrencias (unas 10 o 12) y otra es que se confunden a menudo con usos adjetivales del participio (por ejemplo, soy callada y reservada, la otra era licenciada en administración de empresas, etc. Entre unos y otros había unos 31 casos en el Batch 1 (los primeros 327 clips publicados). Véase la Figura 12.1.

```
821: cosas así saben que siempre <soy reservada> , callada y que hablo
1177: niña siempre ... siempre he <sido callada> y reservada y siempre he
1629: .; Cómo te gustaria «ser recordada> ? Mmm ... como una
2000: que es bueno , que <a sivertido> , uno dice que es
2276: o mexicano , que ya <fue criado> en Estados Unidos . Este
1620! u pina que el español <a href="mailto:sea">sea enseñado> primero ? Pues van ...
17374: Unidos ? Sí , yo <soy nacida> agui . ; Alguna vez
19258: sentido ... aunque mi hermana <a href="mailto:fue nacida>">fue debe ser
26684: sé cómo los escogerían pero <a href="mailto:fue que debe ser
26684: sé cómo los escogerían pero <a href="mailto:fue que debe ser
27681: trabajos siendo bilingúe , que <a href="mailto:fue on la fue ro impresionados">fue nacida>">fue nacida>">fue psicóloga y la otra <a href="mailto:fue on la fue ro impresa">fue on la fue ro impresa inaudible de
46133: es una ciudad muy , <a href="mailto:fue on la fue ro impresa">fue on la fue ro impresa inaudible de
46133: es una ciudad muy , <a href="mailto:fue on la fue ro impresa">fue on la fue ro impresa inaudible de
46133: es una ciudad muy , <a href="mailto:fue on fue on
```

Figure 12.1: Ocurrencias de ser seguido de participio en Batch 1.

Funciones comunicativas

13.1 Expresar deseos con incertidumbre/en el futuro

Esta sección contiene reglas que sirven para detectar estructuras lingüísticas generalmente utilizadas para expresar deseos que implican a uno mismo o a terceros. Esto hace, a menudo y según en qué construcciones, que sea necesario el subjuntivo. Así, por ejemplo, diremos "me gustaría que te casaras conmigo" para hacerle entender a nuestro interlocutor lo que deseamos que realice, pero en cambio diremos "me gustaría casarme conmigo" cuando queremos expresar que es nuestro deseo el que se realice. El primero usa subjuntivo y el segundo no.

13.1.1 Expresar deseos con incertidumbre/en el futuro con agentes distintos en principal y subordinada

Rule

ADDRELATION (Func:Deseos:Ajenos) Verbo IF (0 VerbosExpresarDeseos) (1 Que) TO (*2 Verbo + Subjuntivo BARRIER GrupoNominal OR LimiteOracionSimple OR FiniteVerbForms);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: (...) me sorprendió mi mamá porque no esperaba que me regalaran nada (...)
- DESC: Permite la presencia de clíticos entre el "que" y el subjuntivo que le sigue.

Rule

ADDRELATION (Func:Deseos:Ajenos) Verbo IF (0 VerbosExpresarDeseos) (1 Adverbio OR MejorLiteral LINK 1 Que) TO (*3 Verbo + Subjuntivo BARRIER GrupoNominal OR LimiteOracionSimple OR FiniteVerbForms);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: (...) me gustaría mejor que te casaras conmigo (...)
- DESC: Permite la presencia de adverbios entre el verbo de la principal y el "que".

Rule

ADDRELATION (Func:Deseos:Ajenos) Verbo IF (0 VerbosExpresarDeseos) (1 Que) (2 DetAllIncluded) (3 Sustantivo) (4 PronombresCliticosTODOS OR SeLiteral) TO (*5 Verbo + Subjuntivo BARRIER GrupoNominal OR LimiteOracionSimple OR FiniteVerbForms);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: (...) espero que el español se vuelva tal vez otro idioma oficial (...)
- DESC: Permite la presencia de un sujeto entre el "que" y el subjuntivo.

ADDRELATION (Func:Deseos:Ajenos) OjalaLiteral IF (O OjalaLiteral) (1 Que) (2 DetAllIncluded) (3 Sustantivo) TO (*4 Verbo + Subjuntivo BARRIER GrupoNominal OR LimiteOracionSimple OR FiniteVerbForms);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: Este ... ojalá que algún día cambie , ¿ no ? Pero . ¿ Y tú opinas
- DESC: Permite la presencia de un sintagma nominal sujeto entre el "que" y el subjuntivo.

Rule

ADDRELATION (Func:Deseos:Ajenos) Verbo IF (0 VerbosExpresarDeseos) (1 DetAllIncluded) (2 Sustantivo OR Adj) (3 Que) (4 PronombresCliticosTODOS OR SeLiteral) TO (*5 Verbo + Subjuntivo BARRIER GrupoNominal OR LimiteOracionSimple OR FiniteVerbForms);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: (...) no esperas tu niña que se comporte así (...)
- DESC: Permite un alteración sintáctica entre el sujeto de la subordinada y la posición del "que". Esto es algo inusual en la lengua escrita, pero dejamos la regla por si en el futuro permite detectar construcciones similares en la lengua oral.

Rule

ADDRELATION (Func:Deseos:Ajenos) Sustantivo IF (0
NombresExpresarDeseos) (1 De OR Para) (2 Que) TO (*3 Verbo +
Subjuntivo BARRIER LimiteOracionSimple OR FiniteVerbForms LINK
T:SintagmaNominal-Y-O-Clitico);

- ETIQ: R:Func:Deseos:Ajenos
- EJEM: (...) sin necesidad de que me tengan que pagar.
- DESC: Marca la ocurrencia de nombres como "necesidad, petición" seguidos de un "para" o un "que" y un verbo subjuntivo, y permite que aparezca un clítico o un sintagma nominal (Det N, Det Adj) entre el "que" y el verbo en subjuntivo.

REM: Esta regla utiliza el operador "T:". Esta "T" indica que los que sigue a los dos puntos (:) es un TEMPLATE definido más arriba, en la zona de definición de LIST y SET. Para más detalles véase la documentación vislcg3 y el grupo de debate (GoogleGroup).

13.2 Expresar objetivos y finalidades con "para"

Rule

ADDRELATION (Func:Expresar:Finalidades) Prep IF (0 Para) (1 Verbo + Infinitive) TO (*2 Que BARRIER LimiteOracionSimple OR FiniteVerbForms LINK 1 Verbo + Subjuntivo);

- ETIQ: R:Func:Expresar:Finalidades
- EJEM: (...) Yo crecí pensando que para orar o para pedirle a Dios que hiciera algo (...)
- DESC: XXX
- 13.2.1 Estilo indirecto
- 13.2.2 Expresando opiniones
- 13.3 Expresar opiniones, dudas
- 13.4 Expresar objetivos (con para que)

OBSERVACIÓN: Oración sacada del corpus. 45388: que viene a vivir a ? . ¿ Qué le ¡recomendaría¿ a un hispano que viene a vivir a o a (...) . Esta frase anterior no es la misma construcción (que la de los deseos con sjubjuntivo) porque el que "viene a vivir" es una adjetival de hispano, no una subordinada de recomendar.

OBSERVACIÓN: Oración sacada del corpus. 45410: Estados Unidos?, más como yo . Yo le ¡recomendaría¿, en cuanto al idioma, trate de aprenderlo lo (...) Esta oración refleja la omisión del que (oral o literario, algo recargado). Este tipo de estructuras requerirían un análisis lingüístico más sofisticado, seguramente – sintáctico, análisis de incisos, etc.

13.5 Expresar deseos y recomendaciones (implicando a uno mismo)

Pragmática

14.1 Marcadores discursivos

Véase http://www.cs.famaf.unc.edu.ar/~laura/shallowdisc4summ/index.html.

14.1.1 Marcadores discursivos indicadores de revisión

Rule | ADD (@Prag:MarcadoresDisc:Revision) MarcDiscRevision;

- ETIQ: @:Prag:MarcadoresDisc:Revision
- EJEM: aunque, excepto o pero
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag: MarcadoresDisc: Revision) Sin TO (1 Embargo);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: sin embargo
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag: MarcadoresDisc: Revision) De TO (1 Hecho);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: de hecho
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag:MarcadoresDisc:Revision) En TO (1 Realidad);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: en realidad
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule ADDRELATION (Prag:MarcadoresDisc:Revision) APrep (1 Pesar) TO (2 De);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: a pesar de
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag:MarcadoresDisc:Revision) De (1 Todos) TO (2 Modos);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: de todos modos
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag:MarcadoresDisc:Revision) Es (1 Cierto) TO (2 Que);

- ETIQ: R:Prag:MarcadoresDisc:Revision (relación entre el primer i el último elemento de la locución)
- EJEM: s cierto que
- DESC: [Espacio para describir con palabras lo que hace la regla.]

14.1.2 Marcadores discursivos indicadores de causa

Rule | ADD (@Prag:MarcadoresDisc:Causa) Porque;

- ETIQ: @Prag:MarcadoresDisc:Causa
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule ADDRELATION (Prag:MarcadoresDisc:Causa) Prep IF (0 Por) (1 LoLiteral) TO (2 Tanto);

Rule

ADDRELATION (Prag:MarcadoresDisc:Causa) Prep IF (0 Por) TO (2 Tanto OR FormasDePalabrasEnLocucionesCausalesConPor);

- ETIQ: Asigna @Prag:MarcadoresDisc:Causa.
- EJEM: por lo tanto
- EJEM: por eso
- DESC: Requiere por seguida por LO y TANTO

ADDRELATION (Prag:MarcadoresDisc:Causa) Debido TO (1 APrep);

- ETIQ: R:Prag:MarcadoresDisc:Causa (relación entre el primer i el último elemento de la locución)
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Prag:MarcadoresDisc:Causa) Gracia (O PL) (NOT 2 Dios) TO (1 APrep);

- ETIQ: R:Prag:MarcadoresDisc:Causa (relación entre el primer i el último elemento de la locución)
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule

ADDRELATION (Prag:MarcadoresDisc:Causa) Gracia (O PL) (1 APrep) TO (2 Dios);

- ETIQ: R:Prag:MarcadoresDisc:Causa (relación entre el primer i el último elemento de la locución)
- EJEM: y de ahí fui a la ceremonia y gracias a dios me hice ciudadano
- DESC: [Espacio para describir con palabras lo que hace la regla.]

14.1.3 Marcadores discursivos indicadores de igualdad

Rule

ADD (@Prag:MarcadoresDisc:Igualdad) Adverbio IF (0 Ademas) (NOT 1 De);

Rule ADD (@Prag:MarcadoresDisc:Igualdad) Adverbio IF (0 Tambien);

OBSERVACIÓN: No añado precisamente porque es ambiguo entre su lectura literal (preciso, justo, cierto) como modificador verbal y su lectura como marcador discursivo. Para anotarlo con fiabilidad habría que tener en cuenta su posición respecto al principio de la oración y el verbo principal. Ejemplos: sí muchas pero no sé precisamente cuál cuál sería una, allí vivimos como un año y precisamente yo me tomé me tomaron fotos en un un estudio de fotografía

- ETIQ: @:Prag:MarcadoresDisc:Igualdad
- EJEM: además, también
- DESC: Requiere además o también no seguido de DE

Rule

```
ADDRELATION (Prag:MarcadoresDisc:Igualdad) Prep IF (0 Por) TO (1 Ejemplo);
```

- ETIQ: R:Prag:MarcadoresDisc:Igualdad (relación entre el primer i el último elemento de la locución)
- EJEM: por ejemplo
- DESC: Requiere por seguida por alguna palabra en la lista de palabras predefinidas Lemas-DePalabrasEnLocucionesConPor

Rule | ADDRELATION (Prag:MarcadoresDisc:Igualdad) ConjLetraO TO (1 Sea);

- ETIQ: R:Prag:MarcadoresDisc:Igualdad
- EJEM: o sea
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule ADDRELATION (Prag: MarcadoresDisc: Igualdad) No TO (1 SoloAcento);

- ETIQ: R:Prag:MarcadoresDisc:Igualdad
- EJEM: porque uno *no sólo* puede ayudar en el aspecto de inglés pero en el aspecto del español también
- DESC: [Espacio para describir con palabras lo que hace la regla.]

Rule | ADDRELATION (Prag:MarcadoresDisc:Igualdad) Tal TO (1 Vez);

- ETIQ: R:Prag:MarcadoresDisc:NoTipificado
- EJEM: noto que tal vez sus hijos hablan el spanglish más unos más que otros
- DESC: [Espacio para describir con palabras lo que hace la regla.]

14.1.4 Elementos discursivos usadors para ganar tiempo

Rule ADD (@Prag:RecuDiscurso:GanarTiempo) ArtPronDemo IF (NOT -1 Prep) (0 EsteLiteral) (NOT 1 Nombre);

- ETIQ: @Prag:ElementosDiscurso:GanarTiempo
- EJEM: Este en mi niñez era un poco más soñadora mucho más soñadora de hecho (...)
- DESC: Anota ocurrencies de ESTE que no vayan precedidas de una preposición ni seguidas de un sustantivo.

Estructuras características del habla en hablantes de herencia

15.1 Estructuras derivadas del contacto entre lenguas

15.1.1 Expresiones con cambio de código

Incluímos aquí expresiones que mezclan palabras en ambas lenguas, en este caso el español y el inglés. En la bibliografía especializada se utiliza el término inglés "code switching".

Rule

ADDRELATION (Herit:Contacto:Mixtas) Verbo IF (0 ListaVerbosCorpusParaContactoMixtos) TO (*1 PalabraEnIngles BARRIER PalabraEnEspanol);

- ETIQ: R:Herit:Contacto:Mixtas
- EJEM: hacer request, comer turkey
- DESC: Anota secuencias de un verbo seguid de una o más palabras en inglés. Por ahora lo limitamos a ser, hacer, decir y comer porque TreeTagger anota como verbo algunas palabras que no lo son y no queremos generar ruido para los usuarios.

15.2 Usos de inidicativo dónde se esperaría subjuntivo (en español estándard)

OBSERVACIÓN: Sección en proceso de implementación.

Vocabulario multipalabra

16.1 Nombres propios

16.1.1 Nombres de localizaciones geográficas

```
Rule ADDRELATION (Vocab:Entities:Geo) Articulo IF (0 ("<E1>")) TO (1 ("<Paso>"));
```

```
Rule ADDRELATION (Vocab:Entities:Geo) Nombre IF (0 ("<Estados>")) TO (1 ("<Unidos>"));
```

Rule ADDRELATION (Vocab:Entities:Geo) Nombre IF (0 ("<Ciudad>")) TO (1 ("<Juárez>"));

- ETIQ: R:Vocab:Entities:Geo
- EJEM: Ciudad Juárez, El Paso, Estados Unidos
- DESC: Reglas para anotar entidades frecuentes en el corpus. En este caso de nombres geográficos.

16.1.2 Combinaciones de verbo y preposición de régimen

```
Rule | ADDRELATION (Vocab:Fraseol:VerbPrep) Verbo IF (0 ("hablar")) TO (*1 Prep BARRIER LimiteOracionSimple OR GrupoNominal OR Verbo);
```

Rule ADDRELATION (Vocab:Fraseol:VerbPrep) Verbo IF (0 ("pensar")) TO (*1 Prep BARRIER LimiteOracionSimple OR GrupoNominal OR Verbo);

Rule | ADDRELATION (Vocab:Fraseol:VerbPrep) Verbo IF (0 ("empezar")) TO (*1 Prep BARRIER LimiteOracionSimple OR GrupoNominal OR Verbo);

- ETIQ: R:Vocab:Fraseol:VerbPrep
- EJEM: hablar PREP, pensar PREP, empezar PREP

• DESC: Ahora no se distingue entre usos lexicalizados y usos composicionales (p.e., "empezar en San Antonio y terminar en Austin" se trata igual que "empezé con la tradición en la preparatoria")

